

連載

Mitaka 改造の手引き (4)

～シーケンス 1 : 月の満ち欠け (宇宙) ～

河村聡人 (Mitaka ワーキンググループ)

1. はじめに

当連載は Mitaka (©2005-2023 加藤恒彦, 国立天文台 4 次元デジタル宇宙プロジェクト) のカスタマイズ機能[1]の活用とその普及を目的としています。全 6 回のうち、今回はその 4 回目となります。

連載の予定表

1. 保存と読み込み : 日食 (一地点) [2]
2. コマンドセット 1 : 日食 (多地点) [3]
3. コマンドセット 2 : 星座[4]
4. シーケンス 1 : 月の満ち欠け (宇宙)
5. シーケンス 2 : 月の満ち欠けと日食
6. 字幕と地名 : SPACE-CISTE 簡易版

当連載は Mitaka ワーキンググループ (WG) による講習 (ワークショップ) [5] にて培われた「Mitaka カスタマイズ攻略マニュアル」を下書きとしています。これは Mitaka 同梱の公式マニュアル[6]を補完・要約したものです。その他にも多数のワークショップ向けの教材が、次の QR コード先にあります。



図 1 MitakaWG の Google ドライブ[7]

当連載で扱う内容は Mitaka の利用上の注意[8]に従い、Mitaka の Web サイト[9]にて自由な再配布が認められた範囲です[10]。なお Mitaka のスクリーンショットの掲載には別途許可を取っています。

今回は日時や観測地点などを指定する一連の命令をまとめ一度で実行するコマンドセットを組みました。今回は一連の命令を場面(シーン)ごとに区切り、順次実行していくシーケンスを作成します。

2. 開発環境の確認

当掲載の Mitaka はバージョン 1.74a です。環境構築の詳細は連載の第 1 回と第 2 回を参考にしてください。

3. コマンドセットからシーケンスに

シーケンスは複数のシーンを遷移していくものです。今回はその例として宇宙から見た月の満ち欠けを表示するシーケンスを書きます。新月から始まり、上弦の月、満月、下弦の月、次の新月と時間を進めながら順に解説できるように、都度一時停止をします。

3.1 シーケンス

シーケンスの基本的な形は以下となります。新たなコマンドセットかシーケンスの定義が始まるまで、もしくはファイルの末尾までをひとつのシーケンスの定義と認識します。

```
DefSequence[シーケンス名, 場所]
  KeyString(表示名)
  // 初期状態の命令
  Pause(秒)
  Transition(秒)
  // 次の状態への命令
  Pause(秒)

  AtExit:
  // 終了時の処理
```

コマンドセットとの一番大きな差異は **Pause** と **Transition** の命令です。 **Pause** は一時停止の命令で、指定された秒数（未指定の場合は無限、0 秒も指定可）が経過するか **x** キーを押すことで次の **Pause** までの命令を一気に実行します。この **Pause** までの一区切りがいわばひとつのコマンドセットのようなものです。次の状態へ滑らかに遷移するためには、**Transition** で遷移の秒数を指定します（詳しくは 3.5 節にて）。

またシーケンスは、**AtExit:**以降の行にて、終了時の動作も規定できます。この動作は **z** キーを押してシーケンスを強制終了した場合にも作動します（詳しくは 3.6 節にて）。

3.2 状態の保存（復習）

状態の保存についての詳細は連載の第 1 回をご覧ください。4D2U のロゴやクレジットの表示も忘れずに設定してください。

宇宙モードにて、地球の北極側より見下ろし、月の軌道が画面に収まるまで引きます。日付を調節し、月がおおむね新月になるようにします。また惑星の拡大率をメニューバーの「表示」から「惑星」>「拡大率」>「拡大 2」とします（図 2）。これにより点でしかなかった地球と月がある程度認識可能な大きさになります（図 3）。この状態を **LunarPhase.mcd** として保存します。



図 2 惑星の拡大率の変更

(Mitaka: ©2005-2023 加藤恒彦, 国立天文台 4次元デジタル宇宙プロジェクト)




図 3 保存するシーンのスクリーンショット (Mitaka: ©2005-2023 加藤恒彦, 国立天文台 4次元デジタル宇宙プロジェクト)

3.3 シーケンス化

LunarPhase.mcd ファイルを開いて、内容をシーケンスに書き換えていきます。

1) シーケンスの定義へ書き直す

DefCommandSet を **DefSequence** に、シーケンス名を **STATE** から **LunarPhase** に変更します。また登録先に **MENU_BAR** を指定します。**KeyString** による表示名の設定は任意です。表示名が設定されていない場合、シーケンス名が表示されます。末尾に **Pause()** を追加し、ここまでを初期状態の定義とします。

2) ズームモードの変更を初期状態と分ける

この段階では拡大率を「拡大 2」に設定していますが、シーケンスの起動当初からこの拡大率で表示されると距離と大きさの関係を誤解してしまいそうです。よって、まず「拡大 2」に変更する **SetZoomMode (ZOOM_2)** の **ZOOM_2** を拡大無しの「等倍」である **ZOOM_REAL_SCALE** に変更します。その代わりにファイルの末尾に **SetZoomMode (ZOOM_2)** を追加し、さらに **Pause()** を続けて追加します。これにより正しい大きさと距離とを確実に表示することができます。

ここまでのコードは本稿末尾の A.1.1 となっています。ただし紙幅の都合上、**SwitchSet**

による膨大な数のキーの一括指定の代わりに SwitchOn によって必要十分なキーを設定しています。

3) ここまでの動作を確認する

Mitaka のメニューバーの「ファイル」から「コマンド定義ファイルを開く」を選び、LunarPhase.mcd を開きます。するとメニューバーの「コマンド」以下に「LunarPhase」が追加されます。

「LunarPhase」を選択するとシーケンスが実行されます。まず小さい地球と月が表示されます。x キーを押してシーンを進めると、天体が拡大されます。さらに x キーを 2 度押すとシーケンスを抜けます（1 度目の x キーで末尾の Pause が解除され、2 度目でシーケンスが終了します）。ここではまだ拡大率を戻す処理が未設定なので、シーケンスが終了しても地球や月は拡大されたままです。手動で拡大率を「等倍」に設定しなおしてください。

4) 変化を記述する

これ以後は、変化させる部分だけ記述します。例えばより複雑なシーケンスを組む際に別途保存した状態を利用したいとします。その場合、SwitchSet など重複する部分を再度記述する必要はありません。

次の 3.4 節と 3.5 節において、今回は時間を変化させていきます。

3.4 時間の指定

時刻の設定には SetDateTime が使えますが、汎用性にやや欠けることがあります。ひとつの時間設定を変えると、その他の関連する時間まで設定し直す必要があるからです。

進める時間を指定する AdvanceTime を利用する方が、保守や流用性の観点で優位です。

AdvanceTime (進める時間)

進める時間は実数 + 単位の形をとります。例えば 10sec なら 10 秒、3hour なら 3 時間となります。また単位はイニシャルの一文字、s や h などでも可です。

今回は月相の 4 分の 1 ずつ動かしたいので、大まかな値として 7d つまり 7 日とします。AdvanceTime (7d) を 4 度繰り返し、その間に Pause () を挿入します。最後の AdvanceTime の後に Pause は不要です。シーケンスの最後には自動的に一時停止が入ります。

3.5 遷移の記述

Transition では遷移にかかる総秒数を指定します。今回は遷移時間を 10 秒とするので、各 AdvanceTime の前に Transition (10) を追加します。次の Pause までの範囲ならどのタイミングで Transition を指定しても構いません。前の Pause の直後や直前など、自分のルールを作っておくと良いです。

バージョン 1.7.0 以降では Transition は滑らかな遷移を提供します。これは常に一定の速度で遷移するのではなく、遷移開始からある遷移速度まで一定の加速度で変化し、また遷移終了の際も一定の加速度で減速します (図 4)。




図 4 Transition による状態の滑らかな遷移のイメージ図

このような加速をともなう滑らかな動きは、空間を動く際には酔い防止というメリットがあるのですが、時間の遷移に適用されると誤った印象を与えます。そこで加速の時間 **DT1** と減速の時間 **DT2** をゼロにし、遷移速度を常に一定にします。以下の命令を拡大率の変更を行うタイミング (A.1.1 の 17 行目) に挿入してください。

```
SetTransitionParams (DT1:0)
```

```
SetTransitionParams (DT2:0)
```

DT1、**DT2** および遷移の総秒数の関係性は図 4 にて示しています。これらの設定値により、どのように状態が変化するかは等加速度直線運動の応用問題を解くことでわかります。紙幅の都合上、解法は省略します。

3.6 終了時の処理の設定

AtEnd:の次の行より終了時の処理を記述します。シーケンス内で変更した設定を初期値に戻すと良いです。今回は **DT1**、**DT2** を初期値の **1.0** に、ズームモードを「等倍」の **ZOOM_REAL_SCALE** に設定しなおします。

3.7 動作の確認

これで完成です。A.1.2 のようになっているはずですが、先ほどの動作確認と同様に、**LunarPhase.mcd** を読み込み、シーケンスを実行してみてください。**Mitaka** は読み込んだファイルの更新を自動反映しませんので、ファイルの更新後は再読み込みが必要です。

正しく記述できていれば、地球と月を拡大し、新月から上弦の月、満月、下弦の月、次の新月へと、そのつど一時停止を挟みながらシーンが進みます。

また途中で **z** キーを押すとシーケンスが終了します。最後まで実行した場合も途中で終了した場合も、地球や月の拡大率が「等倍」に戻っていることを確認してください。

4. おわりに

今回紹介したシーケンスを使用することで、様々なメリットが発生します。大きなメリットのひとつが実行する際の操作がタイミングよくキーを押すだけに簡略化されることにより、上映者の負荷が軽減することです。それにより観客により注意を払うことができますし、**Mitaka** に不慣れな方でも上映を試みる事ができます。別の大きなメリットは操作手順の隠蔽です。対象の天体の移動など、通常はメニューバーやスクリーンメニューからの操作が必要な手順がキーを押すだけになります。よりプロらしい上映を行うことができます。

今回はこれまで行ったカスタマイズを総集し、日食に関するシーケンスを作成します。

文 献

- [1] 加藤恒彦 (2019) 「天文ソフト「Mitaka」の最新機能 ～ユーザーによるカスタマイズ機能と教材開発への応用～」第 33 回天文教育研究会収録, pp157-160
- [2] 河村聡人 (2023) 「Mitaka カスタマイズの手引き (1) ～保存と読み込み：日食 (一地点)～」天文教育, vol.35, no.3, pp64-69
- [3] 河村聡人 (2023) 「Mitaka カスタマイズの手引き (2) ～コマンドセット 1：日食 (多地点)～」天文教育, vol.35, no.4, pp28-33
- [4] 河村聡人 (2023) 「Mitaka カスタマイズの手引き (3) ～コマンドセット 2：星座～」天文教育, vol.35, no.5, pp23-28
- [5] 波田野聡美 (2020) 「Mitaka による天文教育／普及ワーキンググループ活動報告」, 第 34 回天文教育研究会収録, pp92-93.
- [6] mitaka 同梱マニュアル：
 - mitaka_manual_J.pdf
 - features_v174.pdf (バージョンに従い名前の一部は変更される)

[7] Mitaka WG の Google ドライブ

<https://drive.google.com/drive/folders/1vPjvW-j1rEZ0AtGC8I5IrreJSL0eVIX?usp=sharing>

[8] Mitaka の利用上の注意

<https://4d2u.nao.ac.jp/policy/>

[9] Mitaka ダウンロードページ

<https://4d2u.nao.ac.jp/mitaka/download/>

[10] 詳しくは[11]にて報告しますが、Mitaka のホームページが更新され、[8][9]の URL も変わっています。また更新に伴う諸問題も現在は復旧しています。これまでと同様にカスタマイズ及びカスタマイズしたファイルの共有ができます。

[11] 河村聡人 (2023) 「Mitaka カスタマイ

ズとゲーミフィケーション」第 37 回天文教育研究会集録 (予定)



河村 聡人

A.1 ソースコード

A.1.1) LunarPhase.mcd をシーケンスにした段階

```

1  DefSequence[LunarPhase, MENU_BAR]
2  SetMainMode(MODE_MAIN_SPACE)
3  SetScale(1.277e-07)
4  SetDateTime(2023/11/13 10:29:60.0 9.0)
5  SetOrientation(-3.849e-01, 1.774e-02, 2.662e-02, 9.224e-01)
6  SetHeadOrientation(1.000e+00, 0.000e+00, 0.000e+00, 0.000e+00)
7  SetViewUpDirection(VIEW_UP_DIRECTION_ZENITH)
8  SwitchOn(
9  SUN_DISP, SUN_NAME, SUN_GLARE, PLANET_DISP, PLANET_NAME, PLANET_ORBIT,
10 PLANET_SELECT, PLANET_DISP_BILLBOARD, PLANET_CLOUD, SATELLITE_DISP,
11 SATELLITE_NAME, SATELLITE_ORBIT, SATELLITE_SELECT, 4D2U_LOGO_DISP,
12 ON_SCREEN_CREDITS_DISP, ENABLE_SHADER, ENABLE_BLACKOUT_SHORTCUT
13 )
14 SetZoomMode(ZOOM_REAL_SCALE)
15 SetTarget(PLNT_EARTH)
16 Pause()
17 SetZoomMode(ZOOM_2)
18 Pause()

```

※ KeyString による表示名設定は任意。設定がない場合はシーケンス名が表示される。

※ 8~14 行目は長大な SwitchSet の代わりに不可欠なキーのみを SwitchOn している。

A.1.2) LunarPhase.mcd (完成形)

```
1 DefSequence[LunarPhase, MENU_BAR]
2   SetMainMode(MODE_MAIN_SPACE)
3   SetScale(1.277e-07)
4   SetDateTime(2023/11/13 10:29:60.0 9.0)
5   SetOrientation(-3.849e-01, 1.774e-02, 2.662e-02, 9.224e-01)
6   SetHeadOrientation(1.000e+00, 0.000e+00, 0.000e+00, 0.000e+00)
7   SetViewUpDirection(VIEW_UP_DIRECTION_ZENITH)
8   SwitchOn(
9     SUN_DISP, SUN_NAME, SUN_GLARE, PLANET_DISP, PLANET_NAME, PLANET_ORBIT,
10    PLANET_SELECT, PLANET_DISP_BILLBOARD, PLANET_CLOUD, SATELLITE_DISP,
11    SATELLITE_NAME, SATELLITE_ORBIT, SATELLITE_SELECT, 4D2U_LOGO_DISP,
12    ON_SCREEN_CREDITS_DISP, ENABLE_SHADER, ENABLE_BLACKOUT_SHORTCUT
13  )
14  SetZoomMode(ZOOM_REAL_SCALE)
15  SetTarget(PLNT_EARTH)
16  Pause()
17  SetTransitionParams(DT1:0)
18  SetTransitionParams(DT2:0)
19  SetZoomMode(ZOOM_2)
20  Pause()
21  Transition(10)
22  AdvanceTime(7d)
23  Pause()
24  Transition(10)
25  AdvanceTime(7d)
26  Pause()
27  Transition(10)
28  AdvanceTime(7d)
29  Pause()
30  Transition(10)
31  AdvanceTime(7d)
32
33  AtExit:
34  SetZoomMode(ZOOM_REAL_SCALE)
35  SetTransitionParams(DT1:1.0)
36  SetTransitionParams(DT2:1.0)
```

※ **KeyString** による表示名設定は任意。設定がない場合はシーケンス名が表示される。

※ 8~14 行目は長大な **SwitchSet** の代わりに不可欠なキーのみを **SwitchOn** している。